
ABSTRACT 

Data comes in all forms, shapes, sizes, complexities, and quality standards. Stored in text-based files, datasets, database 

systems, and other data structures, users across many industries recognize that data is often problematic and plagued with a 

variety of issues, including the absence of identifiers, or "keys", duplicate rows, and lack of standards to name a few. However, 

many data files will require "fine-tuning" the matching process, including the selection of multiple variables such as name, 

email and other social media addresses, telephone number, date of birth, street address, city, state, zip code, and social 

security number to ensure accurate and reliable matches or groups of similar values. This paper addresses the importance of 

working with reliable data sources, but when this is not possible, a comprehensive strategy of data cleaning, data validation, 

and data transformation processes are presented. We emphasize a sampling of data issues that can occur, popular data 

cleaning techniques, user-defined validation methods, data transformation approaches such as traditional merge and join 

techniques, the resolution of key identifier issues, the avoidance and control of false positives and negatives, the introduction 

of different SAS character-handling functions enabling phonetic and text string matching, including the SOUNDEX, SPEDIS, 

COMPLEV, and COMPGED functions, and an assortment and comparison of SAS programming techniques to successfully 

merge/join, match, and cluster less than perfect, or "messy" data. 

 

INTRODUCTION 

When data sources contain consistent and valid data values, share common unique identifier(s), and have no missing 

data, the matching process rarely presents any problems, even when the data is numeric in one dataset and character 

is another dataset. But, when data originating from multiple sources contains duplicate observations, duplicate and/or 

unreliable keys, missing values, invalid values, capitalization and punctuation issues, inconsistent matching variables, 

and imprecise text identifiers, the matching process can be compromised by unreliable and/or unpredictable results or 

the inability to match rows when the keys should match. Users are faced with making judgments about cleaning and 

standardizing data irregularities before attempting to match and process data. To assist in this time-consuming and 

costly process, users frequently turn to using special-purpose programming techniques including the application of 

approximate string matching and/or an assortment of constructive programming techniques to standardize and 

combine datasets together, sometimes basing assumptions on their own knowledge of the data. 

 

DATASETS USED IN EXAMPLES 

This paper illustrates two datasets, Movies_with_Messy_Data and Actors_with_Messy_Data. The 

Movies_with_Messy_Data dataset, illustrated in Figure 1a, contains 31 observations, a data structure of six variables 

with Title, Category, Studio, and Rating defined as character; and Length and Year defined as numeric. Inspection 

shows several data issues in this dataset including the existence of missing data, duplicate observations, spelling 

errors, punctuation inconsistencies, and invalid values. 

 

The Actors_with_Messy_Data dataset, illustrated in Figure 1b, contains 15 observations and a data structure of three 

variables with Title, Actor_Leading and Actor_Supporting defined as character. Inspection shows several data issues 

with this dataset including missing data, spelling errors, punctuation inconsistencies, and invalid values. 

Figure 1b: Actors_with_Messy_Data dataset. 

 
 
 
 
 

Figure 1a: Movies_with_Messy_Data dataset.  
 

 

THE MATCHING PROCESS EXPLAINED 

In an age of endless spreadsheets, publicly available data files, apps, and database management system (DBMS) 

tables, it’s unusual to find a single sheet, data file, table or dataset that contains all the data needed to answer an 

organization’s questions, let alone an accurate and accessible data structure. Today’s data exists in many forms and all 

too often involves matching two or more data sources to create a combined file. The matching process typically 

involves combining two or more datasets, spreadsheets and/or files possessing a shared, common and reliable, 

identifier (or key) to create a single dataset, spreadsheet, and/or data file. The matching process, illustrated in the 

following diagram, shows two tables with a key, Title, to combine the two datasets (tables) together. 

 

MOVIES  ACTORS 

Title  Title 

Length  Actor_Leading 

Category  Actor_Supporting 

Year   

Studio   

Rating   

 

 

But, when a shared and reliable key is associated with input data sources that are nonexistent, inexact, or unreliable, 

the matching process often becomes more involved and problematic. As cited in Sloan and Hoicowitz (2016), special 

processes are needed to successfully match the names, addresses and other content from different files when they 

are similar, but not exactly the same. SAS users have a variety of methods and techniques at their disposal to help 

solve different name matching issues. In the following table, a number of potential matching challenges are illustrated 

when dealing with data sources. 

Matching Challenges 

Phonetic 

Similarity 

Michael  →  Micheal 

Stephen  →  Steven  

Smith  →  Smythe  

Missing Spaces 

& Hyphens 

Mary Ann → MaryAnn 

Mary-Ann → Mary-Anne 

Missing Components 

Mary Frank → Mary Ann Frank 

John Smith → John F. Smith 

Same Name – 

Different Locations 

Joe’s Bar, LA  →  Joe’s Bar, NY 

McDonald’s → MacDonald’s 

Tax Dept (IL) → Tax Dept (AZ) 

Similar or Identical 

Abbreviations 

CA (Canada) → CA (California) 

NY (City) → NY (State) 

Portland (OR) → Portland (ME) 

Numeric 

Representations 

MMDDYYYY→ YYYYMMDD 

Zip codes 02345 → 2345  

SSN 123456789 → 123-45-6789 

Spelling 

Differences 

Honor  →  Honour 

Behavior → Behaviour 

Labor → Labour 

Titles & 

Honorifics 

Mr. → Mister 

Ms. → Miss 

Dr. → Ph.D 

Nicknames 

Bill → William 

Dave → David 

Liz → Elizabeth 

Truncated 

Components 

Ct. → Court 

Ave. → Avenue 

Rd. → Road 

Initials & 

Abbreviations 

J. Smith → John Smith 

Robo → Robo Inc. 

Similar Names 

ABC Co. → ABC Corporation 

Robo LLC → Robo Inc. 

 

 

In a constructive and systematic way, the authors of this paper describe a six-step approach to cleansing data and 

performing fuzzy matching techniques. Parts of the process are run recursively to clean up and match the values in a 

continuous process. 

SIX-STEP FUZZY MATCHING PROCESS 

 

Step 1: 

Determine the Likely Matching Variables. 
 
 

Step 2: 

Understand the Distribution of Data Values. 
 
 
 

Step 3: 

Perform Data Cleaning. 
 
 

Step 4: 

Perform Data Transformations. 
 
 

Step 5: 

Process Exact Matches. 
 
 

Step 6: 

Apply Fuzzy Matching Techniques. 
 

 
 

 

STEP #1: DETERMINE THE LIKELY MATCHING VARIABLES 

In this first step, the names, and attributes (metadata) of likely matching variables are produced. SAS’ CONTENTS 

procedure is specified to produce the names and attributes of each variable to help determine whether any of the 

variables can be used for matching purposes. In addition to producing metadata and reports, SAS PROC CONTENTS can 

also produce output datasets which can be used to automate the initial checks for variables with the same name. 

 

PROC CONTENTS Code: 

PROC CONTENTS DATA=mydata.Movies_with_Messy_Data ; 

RUN ; 

PROC CONTENTS DATA=mydata.Actors_with_Messy_Data ; 

RUN ; 

 

Using the PROC CONTENTS listing, shown in Figure 2, the results of the TITLE variable’s metadata, along with the other 

variables, is produced from both datasets. The Movies_with_Messy_Data dataset’s data structure consists of six 

variables where Title, Category, Studio, and Rating are defined as character variables; and Length (movie length in 

minutes) and Year (year the movie was produced) are defined as numeric variables. The Actors_with_Messy_Data 

dataset’s data structure consists of three character variables: Title, Actor_Leading and Actor_Supporting. 

Results: 

 
 

 

 

 

Figure 2: CONTENTS procedure Output for Movies_with_Messy_Data and Actors_with_Messy_Data datasets. 

 

 

STEP #2: UNDERSTAND THE DISTRIBUTION OF DATA VALUES 

To derive a more accurate picture of the data sources, we suggest that users conduct extensive data analysis by 

identifying missing values, outliers, invalid values, minimum and maximum values, averages, value ranges, duplicate 

observations, distribution of values, and the number of distinct values a categorical variable contains. This important 

step provides an understanding of the data, while leveraging the data cleaning and standardizing activities that will be 

performed later. One of the first things data wranglers will want to do is explore the data using the SAS FREQ 

procedure, or an equivalent approach like Excel Pivot Tables. 

 

PROC FREQ Code: 

PROC FREQ DATA=mydata.Movies_with_Messy_Data ; 

  TABLES _ALL_ / NOCUM NOPERCENT MISSING ; 

RUN ; 

 

 

Reviewing the results, we see an assortment of data issues including “key” values and/or record duplication, data 

accuracy, inconsistent values, missing values, validation, capitalization versus mixed case, data mismatches, and 

incomplete (partial) data issues, as shown in Figure 3. 

Results: 

 

 

 

 

 

 

Figure 3: Distribution of Values from the FREQ Procedure. 

 

Determining the number of distinct values a categorical variable has is critical to the fuzzy matching process. Acquiring 

this information helps everyone involved better understand the number of distinct variable levels, the unique values 

and the number of occurrences for developing data-driven programming constructs and elements. If there are too 

many distinct values for categorical variables, the values can be put into groups using IF-THEN-ELSE, SELECT-WHEN-

OTHERWISE, or CASE logic. The following SAS code provides us with the number of By-group levels for each variable of 

interest we see in Figure 4. 

 

PROC FREQ Code: 

TITLE "By-group NLevels in Movies_with_Messy_Data" ; 

PROC FREQ DATA=mydata.Movies_with_Messy_Data NLEVELS ; 

RUN ; 

 

Results: 

Figure 4:  The number of By-group levels for each variable of interest 

 

STEP #3: PERFORM DATA CLEANING 

Data cleaning, referred to as data scrubbing, is the process of identifying and fixing data quality issues including 

missing values, invalid character and numeric values, outlier values, value ranges, duplicate observations, and other 

anomalies found in datasets. SAS provides many powerful ways to perform data cleaning tasks (Cody, 2017).  

 

Use SAS Functions to Modify Data 

SAS functions are an essential component of the SAS Base software. Representing a variety of built-in and callable 

routines, functions serve as the “work horses” in the SAS software providing users with “ready-to-use” tools designed 

to ease the burden of writing and testing often lengthy and complex code for a variety of programming tasks. The 

advantage of using SAS functions is evidenced by their relative ease of use, and their ability to provide a more 

efficient, robust and scalable approach to simplifying a process or programming task. 

 

It is sometimes necessary to concatenate fields when matching files, because the fields could be concatenated in one 

file while separate in another.  SAS functions span many functional categories, and this paper focuses on those that 

are integral to the fuzzy matching process. The following is a list of alternative methods of concatenating strings 

and/or variables together.  

 

▪ Use the STRIP function to eliminate leading and trailing blanks, and then concatenate the stripped fields using 

the concatenation operator, and insert blanks between the stripped fields. 

▪ Use one of the following CAT functions to concatenate fields: 

 

✓ CAT, the simplest of concatenation functions, joins two or more strings and/or variables together, end-

to-end producing the same results as with the concatenation operator. 

✓ CATQ is similar to the CATX function, but the CATQ function adds quotation marks to any concatenated 

string or variable. 

✓ CATS removes leading and trailing blanks and concatenates two or more strings or variables together. 

✓ CATT removes trailing blanks and concatenates two or more strings and/or variables together. 

✓ CATX, perhaps the most robust CAT function, removes leading and trailing blanks and concatenates 

two or more strings and/or variables together with a delimiter between each. 

 



A General Review of Fuzzy Matching Techniques and Their Capabilities Using SAS®, continued 

 
Page 9 

 

Explore Data Issues with SAS’ PROC FORMAT 

Problems with inaccurately entered data often necessitate time-consuming validation activities. A popular technique 

used by many to identify data issues is to use the FORMAT procedure. In the next example, a user-defined format is 

created with PROC FORMAT, a SAS DATA step identifies data issues associated with the Category variable, and a SAS 

PROC PRINT is specified to display the Category variable’s data issues, which are displayed in Figure 5. 

 

PROC FORMAT, DATA Step and PROC PRINT Code: 

PROC FORMAT LIBRARY=WORK ; 

  VALUE $Category_Validation 

    'Action'               = 'Action' 

    'Action Adventure'     = 'Action Adventure' 

    'Action Cops & Robber' = 'Action Cops & Robber' 

    'Action Sci-Fi'        = 'Action Sci-Fi' 

    'Adventure'            = 'Adventure' 

    'Comedy'               = 'Comedy' 

    'Drama'                = 'Drama' 

    'Drama Mysteries'      = 'Drama Mysteries' 

    'Drama Romance'        = 'Drama Romance' 

    'Drama Suspense'       = 'Drama Suspense' 

    'Horror'               = 'Horror' 

     Other                 = 'ERROR - Invalid Category' 

  /* Other identified categories not listed */ 

  ; 

RUN ; 

 

 

DATA Validate_Category ; 

  SET mydata.Movies_with_Messy_Data ; 

  Check_Category = PUT(Category,$Category_Validation.) ; 

  IF Check_Category = 'ERROR - Invalid Category' THEN 

  DO ; 

     OUTPUT ; 

  END ; 

RUN ; 

 

PROC PRINT DATA=work.Validate_Category 

           NOOBS N ; 

  TITLE "Validation Report for Movie Category Variable" ; 

  VAR Category Title Rating Length Studio Year ; 

RUN ; 

 

Results: 

 
Figure 5:  Validation Report isolating Issues with the Movie Category Variable. 

Once the invalid movie categories are identified with the validation report, users have the option of using one or more 

data cleaning techniques to manually correct, automating the process, or applying fuzzy matching techniques to 

correct (or handle) each invalid movie category. If the values in a dataset are code values, and the values in another 

dataset are the labels for the codes, and a dataset (table) exists that joins the codes to the labels, the CNTLIN feature 

in PROC FORMAT can be used to facilitate the match-merge (or join) between the two datasets. 

 

Add Categories, if Available, to the Start of the Name 

Doing this can eliminate matches that might occur if two businesses in the same general geographic area have the 

same name (for example: Smith’s could describe a hardware store, a restaurant, or another type of business.)  This is 

done in Figure 1, where Category is in the third column. 

 

Remove Special or Extraneous Characters 

Punctuation can differ even when names or titles are the same.  Therefore, we remove the following characters: ‘ “ & 

? – from the movie title. For example, “National Lampoon’s Vacation” and “National Lampoons Vacation” refer to the 

same movie title even though the former contains an apostrophe and the latter does not.  Although the special 

characters can be removed in a number of ways, the next example shows their removal from the Title variable in both 

datasets using the COMPRESS function. The results are displayed in Figure 6. 

 

Code to Remove Special Characters from Title and Perform Matching Process: 

data work.Movies_Cleaned ; 

  set mydata.Movies_with_messy_data ; 

  where title NE ‘’ ; 

  title = 

   compress(Title,"'""&?-") ;/*Remove special chars from Title*/ 

run ; 

 

title "Movies Dataset After Removing Special Characters" ; 

proc print data=work.Movies_Cleaned ; 

run ; 

 

data work.Actors_Cleaned ; 

  set mydata.Actors_with_messy_data ; 

  where title NE ‘’ ; 

  title =  

   compress(Title,"'""&?-") ;/*Remove special chars from Title*/ 

run ; 

 

title "Actors Dataset After Removing Special Characters" ; 

proc print data=work.Actors_Cleaned ; 

run ; 

 

proc sql ; 

 title "Matched Rows from Movies and Actors" ; 

 select DISTINCT M.Title, Rating, Length, Actor_Leading 

  from work.Movies_Cleaned M, 

       work.Actors_Cleaned A 

   where M.Title = A.Title ; 

quit ; 

 

 

Results: 

 Figure 6:  After the Removal of Special Characters and the Results from an Inner Join. 

 

 

Put All Characters in Upper-case Notation and Remove Leading Blanks 

Different data files, datasets, and database tables could have different standards for capitalization, and some 

character strings can be copied in with leading blanks. As found in our example datasets the value contained in the 

Title variable can be stored as all lower-case, upper-case, or in mixed-case which can impact the success of traditional 

merge and join matching techniques. Consequently, to remedy the issue associated with case and leading blanks, we 

recommend using the STRIP function to remove leading and trailing blanks along with the UPCASE function to convert 

all Title values to uppercase characters. For users of other popular programming languages, there is generally an 

equivalent function, or method, available to handle these types of issues. 

 

Remove Words that might or might not Appear in Key Fields 

Commonly used words in language, referred to as stop words, are frequently ignored by many search and retrieval 

processes. Stop words are classified as irrelevant and, as a result, are inserted into stop lists and are ignored. Examples 

include The, .com, Inc, LTD, LLC, DIVISION, CORP, CORPORATION, CO., and COMPANY.  Some data files, datasets, and 

database tables might include these, while others might not. 

 

Choose a Standard for Addresses 

Address fields can present a challenge when analyzing and processing data sources. To help alleviate comparison 

issues, decide whether to use Avenue or Ave, Road or Rd, Street or St, etc, and then convert the address fields 

accordingly or create a user-defined lookup process using PROC FORMAT to match the standard values. 

Rationalize Zip Codes when Matching Addresses, Use Geocodes when Available 

We found it useful to remove the last 4 digits of 9-digit zip codes, because some files might only have 5-digit zip codes.  

Since some files might have zip codes as numeric fields, and other files might have zip codes as character fields, make 

sure to include leading zeroes.  For example, zip codes with a leading zero, as in 08514, would appear in a numeric 

field as 8514 requiring the leading zero to be inserted along with the specification of a Z5. informat and format being 

assigned to the zip code variable. 

 

If working with US zip codes, make sure they are all numeric.  This may not apply for other countries.  One common 

mistake to watch for is that sometimes Canada, with abbreviation CA, is put in as the state CA (California) instead of 

the country CA.  Since Canada has an alphanumeric 6-character zip code, this, hopefully, will be caught when checking 

for numeric zip codes. 

  

If the program has access to geocodes, or if they are in the input data bases, geocodes can provide a further level of 

validation in addition to the zip codes. Base SAS PROC GEOCODE provides several ways to translate addresses or 

geographic locations to geocodes. 

 

Specify the DUPOUT=, NODUPRECS, or NODUPKEYS Options 

A popular and frequently used procedure, PROC SORT, identifies and removes duplicate observations from a dataset. 

By specifying one or more of the SORT procedure’s three options: DUPOUT=, NODUPRECS, and NODUPKEYS, users 

can control how duplicate observations are identified and removed. 

 

PROC SORT’s DUPOUT= option is often used to identify duplicate observations before removing them from a dataset. 

A DUPOUT= option, often specified when a dataset is too large for visual inspection, can be used with the NODUPKEYS 

or NODUPRECS options to name a dataset that contains duplicate keys or entire observations. In the next example, 

the DUPOUT=, OUT= and NODUPKEY options are specified to identify duplicate keys.  The NODUPKEY option removes 

observations that have the same key values, so that only one remains in the output dataset.  The PROC SORT is 

followed by the PROC PRINT procedure so that the results can be examined. 

 

PROC SORT and PROC PRINT Code: 

PROC SORT DATA=mydata.Movies_with_Messy_Data 

        DUPOUT=work.Movies_Dupout_NoDupkey 

          OUT=work.Movies_Sorted_Cleaned_NoDupkey 

        NODUPKEY ; 

  BY Title ; 

  WHERE Title NE “” ; 

RUN ; 

 

PROC PRINT DATA=work.Movies_Dupout_NoDupkey ; 

  TITLE “Observations Slated for Removal” ; 

RUN ; 

 

PROC PRINT DATA=work.Movies_Sorted_Cleaned_NoDupkey ; 

  TITLE “Cleaned Movies Data Set” ; 

RUN ; 

 

The results of the above SAS code are shown in Figure 7. The NODUPKEY option retains only one observation from any 

group of observations with duplicate keys.  When Observations with identical key values are not adjacent to each 

other, users may first need to specify the NODUPKEY or NODUPKEYS option and sort the dataset by all the variables 

(BY _ALL_ ;) to ensure the observations are in the correct order to remove all duplicates (SAS Usage Note 1566, 2000; 

Lafler, 2017). 

Results: 

 

 

 
Figure 7:  Observations Slated for Removal and the Cleaned Movies Dataset. 

 

 

Although the removal of duplicates using PROC SORT is a popular technique among many SAS users, an element of 

care should be given to using this method when processing large datasets. Since sort operations can often be CPU-

intensive, the authors of this paper recommend comparing PROC SORT to procedures like SAS PROC SQL with the 

SELECT DISTINCT keyword and/or SAS PROC SUMMARY with the CLASS statement to determine the performance 

impact of one method versus another. If the dataset is already sorted, using PROC SUMMARY with the BY statement 

will process the data separately and thus more efficiently for each value of the variables in the BY statement. You can 

then use the CLASS statement for the variables you are concerned with when they are not included in the BY 

statement. 

 

STEP #4: PERFORM DATA TRANSFORMATIONS 

Data transformations can be required to compare files. Dataset structures sometimes need to be converted from wide 

to long or long to wide and files may need to be reconciled by having their variables grouped in different ways.  When 

a dataset’s structure and data is transformed, we typically recommend that a new dataset be created from the original 

one. SAS’ PROC TRANSPOSE is handy for restructuring data in a dataset, and is typically used in preparation for special 

types of processing like array processing. In its simplest form, data can be transformed with or without grouping. In 

the next example, the Movies dataset is first sorted in ascending order by the variable RATING then the sorted dataset 

is transposed using the RATING variable as the by-group variable. The result is shown in Figure 8, and it gives all of the 

titles within each rating. 

PROC TRANSPOSE Code: 

PROC SORT DATA=mydata.Movies_with_Messy_Data 

           OUT=work.Movies_Sorted ; 

  BY Rating ; /* BY-Group to Transpose */ 

  WHERE Title NE “” ; 

RUN ; 

 

PROC TRANSPOSE DATA=work.Movies_Sorted 

           OUT=work.Movies_Transposed ; 

  VAR Title ; /* Variable to Transpose */ 

  BY Rating ; /* BY-Group to Transpose */ 

RUN ; 

 

PROC PRINT DATA=work.Movies_Transposed ; 

RUN ; 

 

Results: 

 
Figure 8: Results from Performing a Data Transform with the TRANSPOSE Procedure. 

 

 

STEP 5: PROCESS EXACT MATCHES 

Since we are trying to match entries that do not have an exact match, we can save processing time by immediately 

eliminating the observations (or rows) with missing key information. This can be accomplished in a number of ways, 

including constructing IF-THEN/ELSE or WHERE logic to bypass processing observations with missing movie titles. 

  

Another approach to bypass processing observations with missing movie titles could be to use the NODUP or 

NODUPKEY parameter with SAS’ PROC SORT (more detail on these options will be presented later).  Once observations 

with missing keys are eliminated, the focus can then be turned to processing observations that have exact matches on 

name, address, and as with our example datasets, the Title variable, as shown in Figure 9.  We also process and retain 

the observations that have mismatches on the Title variable, as shown in Figure 10; the observations that did not have 

exact matches on the Title variable from the Movies dataset, as shown in Figure 11; and the observations that did not 

have exact matches on the Title variable from the Actors dataset, as shown in Figure 12. 

 

PROC SORT, DATA Step and PROC PRINT Code: 

proc sort data=mydata.Actors_with_messy_data 

           out=work.Actors_Sorted ; 

 where Title NE "" ; 

 by Title ; 

run ; 

 

proc sort data=mydata.Movies_with_messy_data 

           out=work.Movies_Sorted ; 

 where Title NE "" ; 

 by Title ; 

run ; 

data work.Matches(DROP=Title) 

     work.MisMatches(DROP=Title) 

     work.Movies_with_Unmatched_Obs(KEEP=Title Length Category Year Studio Rating) 

     work.Actors_with_Unmatched_Obs(KEEP=Title Actor_Leading Actor_Supporting) ; 

 merge work.Movies_Sorted (IN=M) 

       work.Actors_Sorted (IN=A) ; 

 by Title ; 

 if M then Title_from_Movies = Title ; 

 if A then Title_from_Actors = Title ; 

 if M and A then output work.Matches ; 

 else if NOT M or NOT A then output work.MisMatches ; 

 if M and NOT A then output work.Movies_with_Unmatched_Obs ; 

 else if A and NOT M then output work.Actors_with_Unmatched_Obs ; 

run ; 

 

proc print data=work.Matches N ; 

 title "Matched Observations with Missing Keys Eliminated" ; 

 var Title_from_Movies Title_from_Actors Length Category Year 

     Studio Rating Actor_Leading Actor_Supporting ; 

run ; 

 

proc print data=work.MisMatches N ; 

 title "MisMatched Observations with Missing Keys Eliminated" ; 

 var Title_from_Movies Title_from_Actors Length Category Year 

     Studio Rating Actor_Leading Actor_Supporting ; 

run ; 

 

proc print data=work.Movies_with_Unmatched_Obs N ; 

 title "Movies with UnMatched Observations" ; 

 var Title Length Category Year Studio Rating ; 

run ; 

 

proc print data=work.Actors_with_Unmatched_Obs N ; 

 title "Actors with UnMatched Observations" ; 

 var Title Actor_Leading Actor_Supporting ; 

run ; 

 

Results: 

 
Figure 9: Matched Observations. 

Figure 10: Mismatched Observations. 

 

 
Figure 11: UnMatched Movies Observations. 

 

 
Figure 12: UnMatched Actors Observations. 

STEP 6: MATCH KEY FIELDS USING FUZZY MATCHING TECHNIQUES 

Once the data has been cleaned and transformed, a variety of fuzzy matching techniques are available for use. As 

mentioned in (Dunn, 2014), these techniques are designed to be used in a systematic way when a reliable key 

between data sources is nonexistent, inexact, or unreliable. 

 

Fuzzy matching techniques are available with most of the leading software languages including SAS, Python, R, Java, 

and others (RosettaCode, 2018). SAS Institute offers four techniques for its users to use with SAS software: the 

SOUNDEX (phonetic matching) algorithm / function, and the SPEDIS, COMPLEV, and COMPGED functions to help make 

fuzzy matching easier and more effective (Lafler and Sloan, 2024, 2023, and 2022; Sloan and Lafler, 2021 and 2018). 

 

Comparison of Fuzzy Matching Techniques 

 
 

APPLY THE SOUNDEX ALGORITHM / FUNCTION 

The Soundex (phonetic matching) algorithm / function involves matching files on words that sound alike. As one of the 

earliest fuzzy matching techniques, Soundex was invented and patented by Margaret K. Odell and Robert C. Russell in 

1918 and 1922, respectively, to help match surnames that sound alike. It is limited to finding phonetic matches and 

adheres to the following rules when performing a search: 

 

▪ Ignores case (case insensitive); 

▪ Ignores embedded blanks and punctuations; 

▪ Is better at finding English-sounding names. 

 

Although the Soundex algorithm / function does a fairly good job with English-sounding names, the authors have 

found that it frequently falls short when dealing with the multitude of data sources found in today’s world economy 

where English- and non-English sounding names are commonplace. It also has been known to miss similar-sounding 

surnames like Rogers and Rodgers while matching dissimilar surnames such as Smith, Smthe and Schmitt (Foley, 

1999). 

 

So, how does the Soundex algorithm / function work? As implemented, SAS determines whether a name (or a 

variable’s contents) sounds like another by converting each word to a code.  The value assigned to the code consists of 

the first letter in the word followed by one or more digits. Vowels, A, E, I, O and U, along with H, W, Y, and non-

alphabetical characters do not receive a coded value and are ignored; and double letters (e.g., ‘TT’) are assigned a 

single code value for both letters. The codes derived from each word conform to the letters and values are found in 

Table 1. 

SOUNDEX Algorithm / Function Rules 

Letter Value 

B, P, F, V 1 

C, S, G, J, K, Q, X, Z 2 

D, T 3 

L 4 

M, N 5 

R 6 

Table 1: Soundex Algorithm / Function Rules 

 

The general syntax of the Soundex algorithm takes the form of: 

 

 Variable =* “character-string” 

 

To examine how the movie title, Rocky, is assigned a value of R22, R has a value of 6 but is retained as R, O is ignored, 

C is assigned a value of 2, K is assigned a value of 2, and Y is ignored. The converted code for “Rocky” is then matched 

with any other name that has the same assigned code. 

 

In the next example, we use the Soundex algorithm’s =* operator in a simple DATA step WHERE statement with the 

work.Movies_with_Unmatched_Obs dataset created in Step #5 earlier, to find similar sounding Movie Titles. 

 

DATA Step Code with SOUNDEX Algorithm: 

DATA work.Soundex_Matches ;  

  SET work.Movies_with_Unmatched_Obs ;  

  WHERE Title =* “Michael” ;  

RUN ; 

 

PROC PRINT DATA=work.Soundex_Matches NOOBS ;  

  TITLE “Soundex Algorithm Matches” ; 

RUN ; 

 

In the next example, the Soundex algorithm is illustrated using the =* operator in a simple SAS PROC SQL step with a 

WHERE-clause to find similar sounding Movie Titles. 

 

PROC SQL Code with SOUNDEX Algorithm: 

proc sql ; 

 select * 

  from work.Movies_with_Unmatched_Obs 

   where Title =* "Michael" ; 

quit ; 

 

The results from both SOUNDEX algorithm examples are displayed in Figure 13. 

 

Results: 

 
Figure 13:  The result of the Soundex match for “Michael” 

The general syntax of the Soundex function takes the form of: 

 

 SOUNDEX(Variable) AS SOUNDEX_Value 

 

In the next example, we use the Soundex function in a simple SQL query with a WHERE-clause to find similar sounding 

Movie Titles to the movie, “Michael” in the work.Movies_with_Messy_Data dataset (or table). 

 

PROC SQL Code with SOUNDEX Function: 

TITLE “Soundex Function Matches” ; 

PROC SQL ; 

  SELECT *, SOUNDEX(Title,“Michael”) AS SOUNDEX_Value 

    FROM mydata.Movies_with_Messy_Data 

      WHERE UPCASE(Title) LIKE “MICH%” ; 

QUIT ; 

 

The results from the SOUNDEX function example are displayed in Figure 14, below. 

 

Results: 

 
Figure 14:  The result of the Soundex function match for “Michael” 

 

 

APPLY THE SPEDIS FUNCTION 

The SPEDIS, or Spelling Distance, function and its two arguments evaluate possible matching scenarios by translating a 

keyword into a query containing the smallest distance value. Because the SPEDIS function evaluates numerous 

scenarios, it can experience varying performance issues in comparison to other matching techniques. The SPEDIS 

function evaluates query and keyword arguments returning non-negative spelling distance values. A derived value of 

zero indicates an exact match. Generally, derived values are less than 100, but, on occasion, can exceed 200. The 

authors have used and recommend using the SPEDIS function to control the matching process by specifying spelling 

distance values greater than zero and in increments of 10 (e.g., 10, 20, etc.). 

 

So, how does the SPEDIS function work? As implemented, the SPEDIS function determines whether two names (or 

variables’ contents) are alike by computing an asymmetric spelling distance between two words. The SPEDIS function 

computes the costs associated with converting the keyword to the query, as illustrated in Table 2. 

 

SPEDIS Cost Rules 

Operation Cost Description 

Match 0 No change 

Singlet 25 Delete one of a double letter 

Doublet 50 Double a letter 

Swap 50 Reverse the order of two consecutive letters 

Truncate 50 Delete a letter from the end 

Append 35 Add a letter to the end 

Delete 50 Delete a letter from the middle 

Insert 100 Insert a letter in the middle 

Replace 100 Replace a letter in the middle 

Firstdel 100 Delete the first letter 

Firstins 200 Insert a letter at the beginning 

Firstrep 200 Replace the first letter 

Table 2: SPEDIS Cost Rules 

The general syntax of the SPEDIS function takes the form of: 

 

 SPEDIS (query, keyword) 

 

In this example, a simple DATA step with a WHERE statement shows the observations derived by the SPEDIS function 

for finding exact matches for the Movie Title, “Michael”. 

 

DATA Step Code with SPEDIS Function: 

PROC PRINT DATA=work.Movies_with_Unmatched_Obs NOOBS ; 

  TITLE "SPEDIS Function Matches" ; 

  WHERE SPEDIS(Title,"Michael") LE 10 ; 

RUN ; 

 

In the next example, a simple PROC SQL query with a WHERE-clause and CALCULATED keyword is specified to capture 

and show the observations derived by the SPEDIS function for finding exact matches for the Movie Title, “Michael”. 

 

PROC SQL Code with SPEDIS Function: 

PROC SQL ; 

 TITLE “SPEDIS Function Matches” ; 

 SELECT *, 

        SPEDIS(Title,“Michael”) AS Spedis_Value 

  FROM work.Movies_with_Unmatched_Obs 

   WHERE CALCULATED Spedis_Value LE 10 ; 

QUIT ; 

 

The results from both SOUNDEX algorithm examples are displayed in Figure 15.  Only “Michael” and “Micheal” were 

chosen.  This matches the result we obtained from the SOUNDEX inquiry displayed in Figure 13. 

 

Results: 

 
Figure 15: The result of a SPEDIS match for “Michael” 

APPLY THE COMPLEV FUNCTION 

The COMPLEV, or Levenshtein Edit Distance, function is another fuzzy matching technique. COMPLEV counts the 

minimum number of single-character insert, delete, or replace operations to determine how close two strings are. 

Unlike the SPEDIS function and COMPGED function (discussed later), the COMPLEV function assigns a score and 

returns a value indicating the number of operations.  The general syntax of the COMPLEV function takes the form of: 

 

COMPLEV ( string-1, string-2 <,cutoff-value> <,modifier> ) 

 

Required Arguments: 

string-1 specifies a character variable, constant or expression. 

string-2 specifies a character variable, constant or expression. 

 

Optional Arguments: 

cutoff-value specifies a numeric variable, constant or expression. If the actual Levenshtein edit distance is 

greater than the value of cutoff, the value that is returned is equal to the value of cutoff. 

 

modifier specifies a value that alters the action of the COMPLEV function. Valid modifier values are: 

▪ i or I Ignores the case in string-1 and string-2. 

▪ l or L Removes leading blanks before comparing the values in string-1 or string-2. 

▪ n or N Ignores quotation marks around string-1 or string-2. 

▪ : (colon) Truncates the longer of string-1 or string-2 to the length of the shorter string. 

 

In Figure 16, below, we show the number of operations that are performed by the COMPLEV function as it compares 

string-1 with string-2. As can be seen, the smaller the LEV_Score the better the match. 

 

 

Figure 16: COMPLEV (Levenshtein Edit Distance) Number of Operations. 

In the example below, we use the COMPLEV function to determine the best possible match with DRAMA. As is shown 

in Figure 17, the COMPLEV_Number column displays the number of operations that have been performed. The lower 

the value the better the match (e.g., 0 = Best match, 1 = Next Best match, etc.).  DRAMA matches itself for a score of 0, 

and DRAMMA is the next best match with a score of 1. 

 

PROC SQL Code with COMPLEV Function: 

proc sql ; 

 select M.Title, 

        Rating, 

        Length, 

        Category, 

        COMPLEV(M.Category,"Drama") AS COMPLEV_Number 

  from work.Movies_with_Unmatched_Obs M 

   order by M.Title ; 

quit ; 

 

Figure 17 shows the calculations for the Levenshtein Edit Distance for different spelling variations for ‘Drama’ in the 

column, CATEGORY. 

 

Results: 

 
Figure 17: The results of a COMPLEV match with the category of Drama. 

 

In the next example, the COMPLEV function’s computed value is limited to 1 or less using a WHERE-clause. The results 

show the observation associated with the movie “The Hunt for Red October” in the string-1 argument matches the 

value of “The Hunt for Red Oktober” in the string-2 argument, as shown in Figure 18. 

 

PROC SQL Code with COMPLEV Function: 

PROC SQL ; 

 SELECT M.Title, 

        A.Title, 

        Rating, 

        Category,  

        Actor_Leading, 

        Actor_Supporting, 

        COMPLEV(M.Title,A.Title) AS COMPLEV_Score 

  FROM work.Movies_with_Unmatched_Obs M, 

       work.Actors_with_Unmatched_Obs A 

   WHERE CALCULATED COMPLEV_Score LE 1 

    ORDER BY M.TITLE ; 

QUIT ; 

 

Results: 

 
Figure 18: The results of a COMPLEV match where the Number of Operations is 1 or less. 

 

In the next example, the COMPLEV function has a modifier value of “INL” to ignore the case, remove leading blanks, 

and ignore quotes around string-1 and string-2 and a value for the COMPLEV_Score of 1 or less. The results show the 

observation associated with the movie “Ghost” in the argument for string-1 matches the value of “GHOST” in the 

argument for string-2, and the observation associated with the movie “The Hunt for Red October” in the string-1 

argument matches the value of “The Hunt for Red Oktober” in the string-2 argument, as shown in Figure 19. 

 

PROC SQL Code with COMPLEV Function and Arguments: 

PROC SQL ; 

 SELECT M.Title, 

        A.Title, 

        Rating, 

        Category,  

        Actor_Leading, 

        Actor_Supporting, 

        COMPLEV(M.Title,A.Title,”INL”) AS COMPLEV_Score 

  FROM work.Movies_with_Unmatched_Obs M, 

       work.Actors_with_Unmatched_Obs A 

   WHERE CALCULATED COMPLEV_Score LE 1 

    ORDER BY M.TITLE ; 

QUIT ; 

 

Results: 

 
Figure 19:  The results of a COMPLEV match with arguments “INL” specified on pairs of titles. 

 

 

APPLY THE COMPGED FUNCTION 

The COMPGED function is another fuzzy matching technique which is facilitated by a SAS function. It works by 

computing and using a Generalized Edit Distance (GED) score when comparing two text strings. The Generalized Edit 

Distance score is a generalization of the Levenshtein edit distance, which is a measure of dissimilarity between two 

strings (Teres, 2011). When using the COMPGED function to match datasets with unreliable identifiers (or keys), the 

higher the GED score the less likely the two strings match (Sloan and Hoicowitz, 2016). Conversely, for the greatest 

likelihood of a match with the COMPGED function users should seek the lowest derived score from evaluating all the 

possible ways of matching string-1 with string-2. 

The COMPGED function returns values that are multiples of 10, e.g., 20, 100, 200, etc. It’s been our experience, as well 

as others, that most COMPGED scores of 100 or less are valid matches for the comparison that they are performing 

(Cadieux and Bretheim, 2014).  The COMPGED function compares two character strings, along with optional 

parameters indicating whether the cases need to match, leading blanks or quotation marks need to be removed, and 

longer strings should be truncated. The general syntax of the COMPGED function takes the form of: 

 

COMPGED ( string-1, string-2 <,cutoff-value> <,modifier> ) 

 

Required Arguments: 

string-1 specifies a character variable, constant or expression. 

string-2 specifies a character variable, constant or expression. 

 

Optional Arguments: 

cutoff-value specifies a numeric variable, constant or expression. If the actual generalized edit distance is 

greater than the value of cutoff, the value that is returned is equal to the value of cutoff. 

 

modifier specifies a value that alters the action of the COMPGED function. Valid modifier values are: 

▪ i or I Ignores the case in string-1 and string-2. 

▪ l or L Removes leading blanks before comparing the values in string-1 or string-2. 

▪ n or N Ignores quotation marks around string-1 or string-2. 

▪ : (colon) Truncates the longer of string-1 or string-2 to the length of the shorter string. 

 

Table 3 shows the different point values that COMPGED assigns for changes from one character string to another. 

 

COMPGED Scoring Algorithm 

Operation 
Default Cost 

in Units 
Description of Operation 

APPEND 50 When the output string is longer than the input string, add any one character to 
the end of the output string without moving the pointer. 

BLANK 10 Do any of the following: 

• Add one space character to the end of the output string without moving 
the pointer. 

• When the character at the pointer is a space character, advance the pointer 
by one position without changing the output string. 

• When the character at the pointer is a space character, add one space 
character to the end of the output string, and advance the pointer by one 
position. 

• If the cost for BLANK is set to zero by the COMPCOST function, the 
COMPGED function removes all space characters from both strings before 
doing the comparison. 

DELETE 100 Advance the pointer by one position without changing the output string. 

DOUBLE 20 Add the character at the pointer to the end of the output string without moving 
the pointer. 

FDELETE 200 When the output string is empty, advance the pointer by one position without 
changing the output string. 

FINSERT 200 When the pointer is in position one, add any one character to the end of the 
output string without moving the pointer. 

FREPLACE 200 When the pointer is in position one and the output string is empty, one character 
is added to the end of the output string, advancing the pointer by one position. 

INSERT 100 Add any character to the end of the output string without moving the pointer. 

MATCH 0 Copy the character at the pointer from the input string to the end of the output 
string, and advance the pointer by one position. 

PUNCTUATION 30 Do any of the following: 

• Add one punctuation character to the end of the output string without 
moving the pointer. 

• When the character at the pointer is a punctuation character, advance the 
pointer by one position without changing the output string. 

• When the character at the pointer is a punctuation character, add one 
punctuation character to the end of the output string, and advance the 
pointer by one position. 

REPLACE 100 Add any one character to the end of the output string, and advance the pointer 
by one position. 

SINGLE 20 When the character at the pointer is the same as the character that follows in 
the input string, advance the pointer by one position without changing the 
output string. 

SWAP 20 Copy the character that follows the pointer from the input string to the output 
string. Then copy the character at the pointer from the input string to the output 
string. Advance the pointer two positions. 

TRUNCATE 10 When the output string is shorter than the input string, advance the pointer by 
one position without changing the output string. 

Table 3: COMPGED scoring algorithm 

 

An example of the scoring used in the SAS COMPGED function when matching string-1 with string-2, re-sorted from an 

example available in the Help screen for the COMPGED function is displayed in Figure 20 (Sloan and Hoicowitz, 2016). 

 

 
Figure 20:  An example of the scoring used while matching on pairs of titles using the COMPGED function. 

In the next example, traditional WHERE-clause logic with the UPCASE function is specified to equate the values of 

string-1 with string-2. Although this approach is less efficient and can be more time consuming than using traditional 

data cleaning methods or the COMPGED function, the results show the value for the movie “Christmas Vacation” in 

the string-1 argument matches the value of “XMAS Vacation” in the string-2 argument, as shown in Figure 21. 

 

PROC SQL Code with Traditional WHERE-clause logic: 

PROC SQL ; 

 SELECT M.Title, 

        A.Title, 

        Rating, 

        Category, 

        Actor_Leading, 

        Actor_Supporting 

  FROM work.Movies_with_Unmatched_Obs M, 

       work.Actors_with_Unmatched_Obs A 

   WHERE UPCASE(A.Title) = "XMAS VACATION" 

     AND UPCASE(M.Title) = "CHRISTMAS VACATION" 

    ORDER BY M.TITLE ; 

QUIT ; 

 

Results: 

 
Figure 21:  The results of using traditional WHERE-clause logic on pairs of titles. 

 

In the next example, the COMPGED function has a “cutoff-value” for the COMPGED_Score set at 100. The results show 

the row associated with the movie “The Hunt for Red October” in the argument for string-1 matches the value of “The 

Hunt for Red Oktober” in the argument for string-2, as shown in Figure 22. 

 

PROC SQL Code with COMPGED Function: 

PROC SQL ; 

 SELECT M.Title, 

        A.Title, 

        Rating, 

        Category,  

        Actor_Leading, 

        Actor_Supporting, 

        COMPGED(M.Title,A.Title) AS COMPGED_Score 

 FROM work.Movies_with_Unmatched_Obs M, 

      work.Actors_with_Unmatched_Obs A 

  WHERE CALCULATED COMPGED_Score LE 100 

   ORDER BY M.TITLE ; 

QUIT ; 

 

Results: 

 
Figure 22:  The results of a COMPGED match on pairs of titles. 

In the next example, the COMPGED function has a modifier value of “INL” to ignore the case, remove leading blanks, 

and ignore quotes around string-1 and string-2 and a “cutoff-value” for the COMPGED_Score set at 100. The results 

show the row associated with the movie “Ghost” in the argument for string-1 matches the value of “GHOST” in the 

argument for string-2, as shown in Figure 23. 

 

PROC SQL Code with COMPGED Function and Arguments: 

PROC SQL ; 

 SELECT M.Title, 

        A.Title, 

        Rating, 

        Category,  

        Actor_Leading, 

        Actor_Supporting, 

        COMPGED(M.Title,A.Title,’INL’) AS COMPGED_Score 

 FROM work.Movies_with_Unmatched_Obs M, 

      work.Actors_with_Unmatched_Obs A 

  WHERE CALCULATED COMPGED_Score LE 100 

   ORDER BY M.TITLE ; 

QUIT ; 

 

Results: 

 
Figure 23:  The results of a COMPGED match with arguments on pairs of titles. 

 

 

 

Summary Of Fuzzy Matching Techniques 

A summary and comparison of fuzzy matching techniques is illustrated in Figure 24, below. 

 

PROC SQL Code with Summary of Fuzzy Matching Techniques: 

PROC SQL ; 

 SELECT Title 

      , Length 

      , Category 

      , Rating 

      , SOUNDEX(Title)           AS SOUNDEX_Value 

      , SPEDIS(Title,'Michael')  AS SPEDIS_Value 

      , COMPLEV(Title,'Michael') AS COMPLEV_Value 

      , COMPGED(Title,'Michael') AS COMPGED_Value 

  FROM MYDATA.Movies_with_Messy_Data 

   WHERE CALCULATED SPEDIS_Value   GE 0 

     AND CALCULATED COMPLEV_Value  GE 0 

     AND CALCULATED COMPGED_Value  GE 0 

    ORDER BY Title ; 

QUIT ; 

Results: 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

Figure 24:  Summary of Fuzzy Matching techniques. 

 

Use the Lower Score 

For those fuzzy matching techniques that are not commutative (it matters which dataset is placed first and which is 

placed second), use the lower score that results from the different sequences. 

 

Eliminate Entries where the Word Counts or Variable Lengths are Significantly Different 

Eliminate entries where the word counts are significantly different (the level of significance will be determined based 

on the datasets being compared). Additionally, eliminate entries where the difference in the length of the character 

variables is significantly different. 

 

VALIDATION 

As can be seen when comparing the SOUNDEX and SPEDIS methods, and when looking at the results of COMPLEV and 

COMPGED, these methods worked well on a test dataset that was designed to illustrate the results. It should be noted 

that the authors found the COMPLEV function to be best used when comparing simple strings where data sizes and/or 

speed of comparison is important, such as when working with large datasets. It should also be noted that generalized 

edit distance computations such as SAS’ COMPGED function requires more processing time to complete due to its 

more exhaustive and thorough capabilities. 

 

Research was conducted on 50,000 business names to manually identify fuzzy matches using SAS’ COMPGED function 

(Sloan and Hoicowitz, 2016). The intent of the study was to identify false negatives by looking at an alphabetic sort of 

the business names. From the extracted test files the authors identified false positives. Finally, the conditions that 

were specified in the COMPGED function were repeated until the false positives and false negatives were significantly 

reduced. This then became part of the fuzzy matching process by efficiently achieving improved results. 

EXTENDING FUZZY MATCHING TECHNIQUES TO NUMERIC VARIABLES 

The examples that we have used so far have focused on matching the values in character variables.  The SAS SOUNDEX 

process and the SAS functions SPEDIS, COMPLEV, and COMPGED all operate on character variables.  However, we can 

run into similar matching issues with numeric variables.  The numeric variables can often be evaluated by fuzzy 

matching techniques when translated to character variables.  Some variables, such as social security numbers, code 

numbers for different variables, numeric address numbers and United States zip codes, value assignments, and value 

rankings can be subject to errors. 

 

Problems with data input errors, different treatments of leading zeroes, different definitions, and confusing match-

merge results can occur with numeric variables.  When comparing measures of academic achievement, for example, 

students and faculty often have different interpretations of final grades (percentage scale of 100, the letter grades A - 

F on a scale of 1-6 for calculation of point averages, normalized grades on a bell curve).  When different transcripts are 

being merged and compared, there can be differences that fuzzy matching could help resolve.  Also, looking at the 

initial code values can sometimes be more accurate than looking at the character values if the differences are due to 

data input errors. 

 

There are also numeric variables that can have issues when derived from character input or translated to character 

output.  A classic example is the input and output of date values.  Some might be input or output as month-day-year 

(common in the United States), some might be input or output as day-month-year (common In Europe), some might 

be input or output as year-month-day (common when storing data for easy sorting), and some do not include the day 

at all (e.g. Aug-2024 or 06/2024). In addition, date values are often input by hand in many environments, and this 

alone can cause errors that need to be detected and rectified.  Huitong Niu and Yan Wang presented a very interesting 

paper at PharmaSUG 2024 on correcting errors in date values that compared the SAS COMPGED function with SAS 

PROC FASTCLUS (“PharmaSUG 2024 - Paper AP- 298 Comparison of Techniques in Merging Longitudinal Datasets with 

Errors on Date Variable: Fuzzy Matching versus Clustering Analysis”). 

 

PROC FASTCLUS is a quick way to sort values into clusters using k-means clustering, which uses an iterative approach 

to assign values to clusters so that each value is in the cluster that minimizes the distance between the value and the 

cluster centroid.  The PROC FASTCLUS statement requires the MAXCLUSTERS= option for the maximum number of 

clusters (the default=100) and/or the RADIUS= option for the minimum distance for selecting new seeds (the 

default=0).  The more clusters we choose, the smaller the distances will be, and the greater chance that we will miss 

some connections but also avoid invalid connections, which is a similar concern to the cutoff value in COMPGED, 

which causes COMPGED to take in more values the larger the cutoff value gets.  In each case, we need to work with 

our clients to determine whether false negatives or false positives are a greater concern.  Niu and Wang’s paper, 

mentioned above, compared PROC FASTCLUS and COMPGED in terms of how well they corrected different types of 

date value errors in longitudinal studies for medical research. 

 

CONCLUSION 

When data originating from multiple sources contain duplicate observations, duplicate and/or unreliable keys, missing 

values, invalid values, capitalization and punctuation issues, inconsistent matching variables, and imprecise text 

identifiers, the matching process is often compromised by unreliable and/or unpredictable results. This paper 

demonstrates a five-step approach including identifying, cleaning and standardizing data irregularities, conducting 

data transformations, and utilizing special-purpose programming techniques such as the application of SAS functions, 

the SOUNDEX algorithm and function, the SPEDIS function, approximate string matching functions including COMPLEV 

and COMPGED, and an assortment of constructive programming techniques to standardize and combine datasets 

together when the matching columns are unreliable or less than perfect. 

